Jour. Fac. Sci., Hokkaido Univ., Ser. VII (Geophysics), Vol. 11, No. 4, 739-751, 2000.

A Computer Program for the Calculation of Piezomagnetic Field due to a Spherical Pressure Source (Mogi Model) in the Inhomogeneously Magnetized Crust

Mitsuru Utsugi

Division of Earth and Planetary Sciences, Graduate School of Science,
Hokkaido University, Sapporo 060-0810, Japan

(Received November 30, 1999)

Abstract

We developed an analytical method for evaluating the piezomagnetic field considering the inhomogeneously magnetized crust. In this paper, we present a FORTRAN program to calculate the piezomagnetic effect due to a spherical pressure source (the Mogi model) in the inhomogeneously magnetized crust.

1. Introduction

In the previous studies (Sasai, 1991; Utsugi et al.,1999), the analytical solution of piezomagnetic effect has been obtained in the case of uniformly magnetized crustal model. However, it is self-evident that the earth's crust is inhomogeneously magnetized. Then, to represent the inhomogeneity of the crustal magnetization, we divide the crust into a number of compartments. Each compartment is assumed to have its own uniform magnetic properties such as the magnetization and the stress sensitivity. The geomagnetic field change at a certain point on the earth's surface may be approximated by the sum of the piezomagnetic field derived from each compartment. The piezomagnetic field is generally expressed by the surface integral of displacement and its derivation over the boundary surface of magnetized region. In the case of the compartment model, the surface integral becomes finite one. This integral cannot be solved analytically. However the line integral with respect to either coordinate can be represented analytically using elliptic integrals. There are several simple algorithms to make rapid and exact evaluation of the elliptic integrals. Using these algorithms, the piezomagnetic field can be expressed by line integral. Through numerical evaluations of this line integral by using the double exponential method (e.g. Takahashi and Mori, 1974), we can obtain the piezomagnetic field considering the inhomogeneity of the crustal magnetization.

In this paper, we present a FORTRAN program which calculates the piezomagnetic effect due to a spherical pressure source (the Mogi model: Mogi, 1958) in the inhomogeneously magnetized crust.

2. Geomagnetic field changes due to Mogi model

We consider the coordinate system as shown in Fig. 1. A semi-infinite elastic medium occupies $z>0$. We assume that the region V_1 is uniformly magnetized cube and outside V_2 is demagnetized. We also assume that the elastic properties such as Lame constants λ, μ are common in the regions V_1 and V_2. The analytical solution of the displacement $\mathbf{u}$ due to the Mogi model is obtained by Mindlin and Cheng (1950) and Yamakawa (1955) as follows:

$$u_x=\frac{C}{2\mu}\left[\frac{x}{R_1^3}+2\frac{x}{R_2^3}-\frac{6xz(z+D)}{R_2^5}\right] \tag{1}$$

$$u_y=\frac{C}{2\mu}\left[\frac{y}{R_1^3}+2\frac{y}{R_2^3}-\frac{6yz(z+D)}{R_2^5}\right] \tag{2}$$

$$u_z=\frac{C}{2\mu}\left[\frac{z-D}{R_1^3}-2\frac{D}{R_2^3}-\frac{6z(z+D)^2}{R_2^5}\right] \tag{3}$$

where $R_1=\sqrt{x^2+y^2+(z-D)^2}$, $R^2=\sqrt{x^2+y^2+(z+D)^2}$ and D is the source depth. The moment C is given by



Fig. 1. Coordinate system, source and crustal model are shown. A semi-infinite elastic medium occupies $z>0$. The cubic region V_1 ($L\times W\times H$ km^3) is uniformly magnetized and outside V_2 is demagnetized. In this medium, a spherical pressure source (the Mogi model) is assumed.

$$C=-\frac{1}{2}a^3\varDelta P,$$

where a is the radius of the sphere and $\varDelta P$ is the hydrostatic pressure acting on the surface of the sphere.

According to the representation theorem for the piezomagnetic field (Sasai, 1991), the geomagnetic change $\varDelta\mathbf{M}$ is expressed by the integral over the boundary surface S_1 (Fig. 1) of the cubic region:

$$\begin{aligned}\varDelta\mathbf{M}^k(\mathbf{r}_0)=&-C_k\iint_{S_1}\left[\left\{-\frac{\partial u_k(\mathbf{r})}{\partial n}+\frac{2(\lambda+\mu)}{3\lambda+2\mu}\varDelta\mathbf{m}^k\cdot\mathbf{n}\right\}\frac{\partial}{\partial\mathbf{r}_0}\frac{1}{\rho}\right.\\&\left.+u_k(\mathbf{r})\frac{\partial^2}{\partial\mathbf{r}_0\partial n}\left(\frac{1}{\rho}\right)\right]dS_{\mathbf{r}},\\C_k=&\frac{1}{2}\beta\mu\frac{3\lambda+2\mu}{\lambda+\mu}J_k,\\\varDelta m_i^k=&\frac{3}{2}\left(\frac{\partial u_k}{\partial x_i}+\frac{\partial u_i}{\partial x_k}\right)-\delta_{ij}\mathrm{div}\mathbf{u},\\\rho=&|\mathbf{r}_0-\mathbf{r}|,\end{aligned}\tag{4}$$

where δ_{ij} is the Kronecker's delta, J_k is the k-th component of initial magnetization within V_1 and $\mathbf{u}$ is the displacement vector. $\mathbf{r}_0$ and $\mathbf{r}$ indicate the observation point and the arbitrary point within the medium, respectively.

Substituting eqs. (1) to (3) into eq. (4), the geomagnetic field changes are written in the form of surface integral of function $f(1/R, 1/\rho)$. In the present case, we have to make the finite surface integral because S_1 is finite. As mentioned in the previous chapter, we cannot solve this finite surface integral analytically. However, using elliptic integrals, the integral with respect to either coordinate can be represented by the analytical form. Then we can transform the surface integral of eq. (4) to the line integral as shown in the following chapter.

3. Line integrals of $f(1/R, 1/\rho)$

In eq. (4), we see the following integrals of function $f(1/R, 1/\rho)$:

$$\varPhi_{ij}(y, z, x_0, y_0, z_0\,;\ D)=\int\frac{1}{R^i\rho^j}dx,\tag{5}$$

$$\varPsi_{ij}(y, z, x_0, y_0, z_0\,;\ D)=\int\frac{x}{R^i\rho^j}dx,\tag{6}$$

These integrals can be solved analytically using the following elliptic integrals:

$$F(\varphi, k)=\int_0^{\varphi}\frac{1}{\sqrt{1-k^2\sin^2\varphi'}}d\varphi'\quad\text{(First Kind)},\tag{7}$$

$$E(\varphi, k)=\int_0^{\varphi}\sqrt{1-k^2\sin^2\varphi'}\,d\varphi' \quad \text{(Second Kind)}. \tag{8}$$

For example, we consider the following integral:

$$\Phi_{11}=\int\frac{1}{R\rho}dx. \tag{9}$$

For the convenience, we denote R and ρ as follows:

$$\begin{cases} R^2=x^2+\zeta^2, \\ \rho^2=x^2-2x_0x+c_0^2, \\ \zeta^2=y^2+(z-D)^2, \\ c_0^2=x_0^2+c^2, \\ c^2=(y_0-y)^2+(z_0-z)^2, \end{cases}$$

where (x, y, z) is the arbitrary point within the magnetized region of the medium, and (x_0, y_0, z_0) is the observation point.

To express eq. (9) using the elliptic integrals, we introduce the following variable transform:

$$x\rightarrow\frac{\alpha t+\beta}{t+1}, \tag{10}$$

where

$$\begin{cases} \alpha=\dfrac{c_0^2-\zeta^2-D}{2x_0}, \\ \beta=\dfrac{c_0^2-\zeta^2+D}{2x_0} \quad (\alpha\leq\beta), \\ D^2=(c_0^2-\zeta)^2+4x_0^2\zeta^2. \end{cases}$$

Using eq. (10), R and ρ become more simply:

$$R^2=\frac{1}{(t+1)^2}\{(\alpha^2+\zeta^2)t^2+(\beta^2+\zeta^2)\},$$

$$\rho^2=\frac{1}{(t+1)^2}\{(\alpha^2-2x_0\alpha+c_0^2)t^2+(\beta^2-2x_0\beta+c_0^2).$$

Therefore the terms $1/R$ and $1/\rho$ are written as follows:

$$\frac{1}{R}\rightarrow\frac{1}{\sqrt{p(\alpha)}}\frac{|t+1|}{\sqrt{t^2+\xi^2}}, \quad \frac{1}{\rho}\rightarrow\frac{1}{\sqrt{q(\alpha)}}\frac{|t+1|}{\sqrt{t^2+\eta^2}},$$

where

$$\begin{cases} p(s)=s^2+\zeta^2, \\ q(s)=(x_0-s)^2+c^2, \\ \xi^2=p(\beta)/p(\alpha), \\ \eta^2=q(\beta)/q(\alpha) \quad (\xi^2>\eta^2), \end{cases}$$

and eq. (9) is rewritten as follows:

$$\Phi_{11}=\frac{\alpha-\beta}{\sqrt{p(\alpha)q(\alpha)}}\int\frac{1}{\sqrt{t^2+\xi^2}\sqrt{t^2+\eta^2}}dt. \tag{11}$$

Transforming $t\rightarrow\varphi=\tan^{-1}\left(\frac{t}{\eta}\right)$, we obtain the following result:

$$\Phi_{11}=\frac{\alpha-\beta}{\sqrt{p(\alpha)q(\alpha)}}\frac{1}{\xi}F(\varphi, k),$$

where $k^2=(\xi^2-\eta^2)/\xi^2$. To evaluate the elliptic integrals, there are some algorithms (Cayley, 1961; Byrd and Friedman, 1954) without solving eqs. (7) and (8) directly. Using these algorithms, we can evaluate Φ_{11} easily. With the same manner, Φ_{ij} and Ψ_{ij} are solved analytically. The exact forms of Φ_{ij} and Ψ_{ij} are given by Utsugi (1999).

The geomagnetic change is written by the following line integral of $\mathbf{g}$ which is an arbitrary function of Φ_{ij} and Ψ_{ij}:

$$\Delta\mathbf{M}^k(\mathbf{r}_0)=C_k\int\mathbf{g}(\Phi_{ij}(\mathbf{r}_0, \mathbf{r}), \Psi_{ij}(\mathbf{r}_0, \mathbf{r}))dl_{\mathbf{r}}. \tag{12}$$

From numerical calculation of this integral, we can evaluate the piezomagnetic change considering the inhomogeneously magnetized crust. To evaluate eq. (12), we use the double exponential integral method (DEM).

4. Programs

The source list of the programs for calculating the piezomagnetic field is given in Appendix. The subroutine 'MGINHOMO' calculates the geomagnetic

Table 1. Input parameters.

X0, Y0, Z0	x_0, y_0, z_0	(km)	Observation point
CX0, CY0, CZ0		(km)	Location of the center of the cube
CL, CW, CH	L, W, H	(km)	Length, width and height of the cube
C0	$a^3\Delta P/2$	(km^3·bar)	Moment of Mogi model
D0	D	(km)	Source depth
AMU	μ	(cgs)	Rigidity
POI	v		Poisson ratio
CMZX, CMZY, CMZZ	$\mathbf{J}=(J_x, J_y, J_x)$	(A/m)	Magnetization vector within the region V_1
BETA	β	(bar^{-1})	Stress sensitivity

change due to a magnetized cubic block. This subroutine requires the parameters as shown in Table 1 and returns eastward (DMX), northward (DMY), downward (DMZ) and total force (DMF) components of geomagnetic change.

The double precision functions 'MGXYX_i', 'MGXZX_i' and 'MGYZX_i' (X_i= X, Y or Z) calculate the contributions from the $x-y$, $x-z$ and $y-z$ plane of V_1, respectively. The subroutine 'PHSIIJ' calculates Φ_{ij} and Ψ_{ij}. The elliptic integrals which appear in Φ_{ij} and Ψ_{ij} are calculated by the subroutine 'ELLIPFE'. The subroutine 'DEMINT' calculates the numerical line integral of the functions of Φ_{ij} and Ψ_{ij} numerically using DEM.

5. Numerical example

In Fig. 2, we show a numerical example. This figure shows the profiles of the total force of geomagnetic change along the y axis ($x_0=0$, $z_0=1$)(m). The case A in Fig. 2 is based on the uniformly magnetized crust: a layer $0<z<H_c$ $=15$ km (H_c indicates the Curie depth) is uniformly magnetized by 1 A/m. The case B is based on the inhomogeneously magnetized crust as shown in Fig. 1. The intensity of magnetization within the cube is assumed as 1 A/m. In both



Fig. 2. Profiles of the total force of geomagnetic change along the y axis ($x_0=0$, $z_0=1$ m). Cases A and B indicate the geomagnetic changes based on an uniformly magnetized crustal model and a cubic model, respectively.

Table 2. Cubic model parameters.

(X0, Z0)	(0, 1)	(m)
(CX0, CY0, CZ0)	(10, 0, 7.5)	(km)
(CL, CW, CH)	(10, 10, 15)	(km)
C0	10^3	(km^3·bar)
D0	5	(km)
AMU	3.5×10^{11}	(cgs)
POI	0.25	
(CMZX, CMZY, CMZZ)	$(0, 1/\sqrt{2}, 1/\sqrt{2})$	(A/m)
BETA	10^{-4}	(bar^{-1})

cases, the magnetic inclination and declination are assumed as 45° and 0°, respectively. The model parameters are given in Table 2. Through this calculations, it becomes clear that the piezomagnetic effect is enhanced around the edges of the cubic block (at $y_0=5$ and $y_0=15$ km). This is caused by the fact that, unlike a uniform medium, the magnetic fields arising from stress-induced magnetic dipoles do not cancel with one another around the edges. The mechanism of this enhancement is well discussed in Oshiman (1990) and Utsugi (1999).

References

Byrd, P.F. and M.D. Friedman, 1954. *Handbook of elliptic integrals for engineers and physicists*, Springer Verlag, pp. 335.

Cayley, A., 1961. *An elementary treatise on elliptic functions*, Dover Publications, Inc, pp. 386.

Mindlin, R.D. and D.H. Cheng, 1950. Nuclei of strain in the semi-infinite solid. J. Applied. Phys, **21**, 926-930.

Mogi, K., 1958. Relations between the eruptions of various volcanoes and the deformations of the ground surfaces around them. Bull. Earthq. Res. Inst., Univ. Tokyo, **36**, 99-134.

Oshiman, N., 1990. Enhancement of tectonomagnetic change due to non-uniform magnetization in the Earth's crust -two dimensional case studies. J. Geomag. Geoelectr., **42**, 607-619.

Sasai, Y., 1991. Tectonomagnetic modeling on the basis of linear piezomagnetic effect. Bull. Earthq. Res. Inst., Univ. Tokyo, **66**, 585-722.

Takahashi, R. and M. Mori, 1974. Double exponential formulas for numerical integration, Publ. R.I.M.S., Kyoto Univ., **9**, 721-741.

Utsugi, M., 1999. A theoretical study on seismomagnetic effect considering the inhomogeneously magnetized Earth's crust. Ph.D. Thesis, Hokkaido Univ., pp. 124.

Utsugi, M., Y. Nishida, and Y. Sasai, 1999. Piezomagnetic potentials due to an inclined rectangular fault in a semi-infinite medium. Geophys. J. Int., **133** (in press).

Yamakawa, N., 1955. On the strain produced in a semi-infinite elastic solid by an interior source of stress. J. Seism. Soc. Japan, (II), 8, 84-98 (in Japanese with English abstract).

```fortran
      SUBROUTINE MGINHOMO(X0, Y0, Z0, CX0, CY0, CZ0,
     &     CL, CW, CH, C0, D0, AMU, POI, CMZX, CMZY, CMZZ,
     &     DMX, DMY, DMZ, DMF)
C**********************************************************************
C
C      GEOMAGNETIC FIELD CHANGE AT OUTSIDE THE MEDIUM
C                 DUE TO THE MOGI MODEL
C               BASED ON THE CUBIC BLOCK MODEL
C              CODED BY M. UTSUGI....DEC 1999
C
C***********************************************************************
C**** INPUT PARAMETERS
C     X0, Y0, Z0        : LOCATION OF THE OBSERVATION POINT
C     CX0, CY0, CZ0     : LOCATION OF CENTER OF CUBIC BLOCK
C     CL, CW, CH        : LENGTH, WIDTH AND HEIGHT OF CUBE
C     C0                : MOMENT OF MOGI NODEL
C     D0                : SOURCE DEPTH
C     AMU               : RIGIDITY
C     POI               : POISSON RATIO
C     CMZX, CMZY, CMZZ  : MAGNETIZATION GAP THROUGH THE BOUNDARY
C                         SURFACE OF CUBE
C     BETA              : STRESS SENSITIVITY
C
C***** OUTPUT VALUES
C     DMX,DMY,DMZ,DMF : EAST, NORTH, DOWNWARD COMPONENTS
C                       AND TOTAL FORCE OF GEOMAGNETIC
C                       FIELD CHANGE (NT)
C
      IMPLICIT REAL*8 (A-H, O-Z)
      INTEGER NEND, NPOW
      COMMON /COM1/ NEND, NPOW, AO(2),
     &     AP(608, 2), AM(608,2), BO, BB(608)
      COMMON /COM2/ C00, C0X, C0Y, C0Z
C
      EXTERNAL MGXYX, MGXYY, MGXYZ
      EXTERNAL MGXZX, MGXZY, MGXZZ
      EXTERNAL MGYZX, MGYZY, MGYZZ
C
      EPS = 1.D-15
      L = 0
      CMZ0 = SQRT(CMZX**2+CMZY**2+CMZZ**2)
      SMM=CMZ0*BETA*AMU*(1.D0+POI)
      C00=0.5D0*SMM*1.D-7
      C0X=CMZX*C00
      C0Y=-CMZY*C00
      C0Z=CMZZ*C00
C
      CALL FIAB (NEND, NPOW, AO, AM, AP, BO, BB)
C
C***  DMXY  : CONTRIBUTIONS FROM X-Y PLANE  ***
C
C     DMXYX : EASTWARD COMPONENTS
C
      CALL DEMINT (MGXYX, X0, Y0, Z0, D0, CZ-0.5D0*CH,
     &     CX-0.5D0*CL, CX+0.5D0*CL,
     &     CY-0.5D0*CW, CY+0.5D0*CW,
     &     EPS, L, DMXYX0)
      CALL DEMINT (MGXYX, X0, Y0, Z0, D0, CZ+0.5D0*CH,
     &     CX-0.5D0*CL, CX+0.5D0*CL,
     &     CY-0.5D0*CW, CY+0.5D0*CW,
     &     EPS, L, DMXYXH)
C
      DMXYX=DMXYX0-DMXYXH
C
C     DMXYY: NORTHWARD COMPONENTS
C
      CALL DEMINT (MGXYY, X0, Y0, Z0, D0, CZ-0.5D0*CH,
     &     CX-0.5D0*CL, CX+0.5D0*CL,
     &     CY-0.5D0*CW, CY+0.5D0*CW,
     &     EPS, L, DMXYY0)
      CALL DEMINT (MGXYY, X0, Y0, Z0, D0, CZ+0.5D0*CH,
     &     CX-0.5D0*CL, CX+0.5D0*CL,
     &     CY-0.5D0*CW, CY+0.5D0*CW,
     &     EPS, L, DMXYYH)
C
      DMXYY=DMXYY0-DMXYYH
C
C     DMXYZ: DOWNWARD COMPONENTS
C
      CALL DEMINT (MGXYZ, X0, Y0, Z0, D0, CZ-0.5D0*CH,
     &     CX-0.5D0*CL, CX+0.5D0*CL,
     &     CY-0.5D0*CW, CY+0.5D0*CW,
     &     EPS, L, DMXYY0)
      CALL DEMINT (MGXYZ, X0, Y0, Z0, D0, CZ+0.5D0*CH,
     &     CX-0.5D0*CL, CX+0.5D0*CL,
     &     CY-0.5D0*CW, CY+0.5D0*CW,
     &     EPS, L, DMXYYH)
C
      DMXYY=DMXYY0-DMXYYH
C
C***  DMXZ  : CONTRIBUTIONS FROM X-Z PLANE  ***
C
C     DMXZX : EASTWARD COMPONENTS
C
      CALL DEMINT (MGXZX, X0, Y0, Z0, D0, CY-0.5D0*CW,
     &     CX-0.5D0*CL, CX+0.5D0*CL,
     &     CZ-0.5D0*CH, CZ+0.5D0*CH,
     &     EPS, L, DMXZX0)
      CALL DEMINT (MGXZX, X0, Y0, Z0, D0, CY+0.5D0*CW,
     &     CX-0.5D0*CL, CX+0.5D0*CL,
     &     CZ-0.5D0*CH, CZ+0.5D0*CH,
     &     EPS, L, DMXZXH)
C
      DMXZX=DMXZX0-DMXZXH
C
C     DMXZY: NORTHWARD COMPONENTS
C
      CALL DEMINT (MGXZY, X0, Y0, Z0, D0, CY-0.5D0*CW,
     &     CX-0.5D0*CL, CX+0.5D0*CL,
     &     CZ-0.5D0*CH, CZ+0.5D0*CH,
     &     EPS, L, DMXZY0)
      CALL DEMINT (MGXZY, X0, Y0, Z0, D0, CY+0.5D0*CW,
     &     CX-0.5D0*CL, CX+0.5D0*CL,
     &     CZ-0.5D0*CH, CZ+0.5D0*CH,
     &     EPS, L, DMXZYH)
C
      DMXZY=DMXZY0-DMXZYH
C
C     DMXZZ: DOWNWARD COMPONENTS
C
      CALL DEMINT (MGXZZ, X0, Y0, Z0, D0, CY-0.5D0*CW,
     &     CX-0.5D0*CL, CX+0.5D0*CL,
     &     CZ-0.5D0*CH, CY+0.5D0*CW,
     &     EPS, L, DMXZZ0)
      CALL DEMINT (MGXZZ, X0, Y0, Z0, D0, CY+0.5D0*CW,
     &     CX-0.5D0*CL, CX+0.5D0*CL,
     &     CZ-0.5D0*CH, CY+0.5D0*CW,
     &     EPS, L, DMXZZH)
C
      DMXZZ=DMXZZ0-DMXZZH
C
C***  DMYZ  : CONTRIBUTIONS FROM Y-Z PLANE  ***
C
C     DMYZX : EASTWARD COMPONENTS
C
      CALL DEMINT (MGYZX, X0, Y0, Z0, D0, CX-0.5D0*CL,
     &     CY-0.5D0*CW, CY+0.5D0*CW,
     &     CZ-0.5D0*CH, CZ+0.5D0*CH,
```

```
     &     EPS, L, DMYZX0)
      CALL DEMINT (MGYZX, X0, Y0, Z0, D0, CX+0.5D0*CL,
     &     CY-0.5D0*CW, CY+0.5D0*CW,
     &     CZ-0.5D0*CH, CZ+0.5D0*CH,
     &     EPS, L, DMYZXH)
C
      DMYZX=DMYZX0-DMYZXH
C
C     DMYZY: NORTHWARD COMPONENTS
C
      CALL DEMINT (MGYZY, X0, Y0, Z0, D0, CX-0.5D0*CL,
     &     CY-0.5D0*CW, CY+0.5D0*CW,
     &     CZ-0.5D0*CH, CZ+0.5D0*CH,
     &     EPS, L, DMYZY0)
      CALL DEMINT (MGYZY, X0, Y0, Z0, D0, CX+0.5D0*CL,
     &     CY-0.5D0*CW, CY+0.5D0*CW,
     &     CZ-0.5D0*CH, CZ+0.5D0*CH,
     &     EPS, L, DMYZYH)
C
      DMYZY=DMYZY0-DMYZYH
C
C     DMYZZ: DOWNWARD COMPONENTS
C
      CALL DEMINT (MGYZZ, X0, Y0, Z0, D0, CX-0.5D0*CL,
     &     CY-0.5D0*CW, CY+0.5D0*CW,
     &     CZ-0.5D0*CH, CY+0.5D0*CW,
     &     EPS, L, DMYZZ0)
      CALL DEMINT (MGYZZ, X0, Y0, Z0, D0, CX+0.5D0*CL,
     &     CY-0.5D0*CW, CY+0.5D0*CW,
     &     CZ-0.5D0*CH, CY+0.5D0*CW,
     &     EPS, L, DMYZZH)
C
      DMYZZ=DMYZZ0-DMYZZH
C
      DMX=DMXYX+DMXZX+DMYZX
      DMY=DMXYY+DMXZY+DMYZY
      DMZ=DMXYZ+DMXZZ+DMYZZ
C
      DMF=(DMX*CMZX+DMY*CMZY+DMZ*CMZZ)/CMZ0
C
      RETURN
      END
C
C
C
      SUBROUTINE DEMINT (FUNC, X0, Y0, Z0, D0,
     &     Z, X1, X2, A, B, EPS, L, V)
C
      IMPLICIT REAL*8 (A-H, O-Z)
      INTEGER NEND, NPOW
      COMMON /COM1/ NEND, NPOW, AO(2),
     &     AP(608, 2), AM(608,2), BO, BB(608)
C
      DATA HALF, EPSO / 0.5D0, 1.0D-32 /
      DATA EPSM, EPSP / 0.D0, 0.D0 /
C
      FAC = (B - A) * HALF
      IF (L .EQ. 0) THEN
         SHFM = (B + A) * HALF
         SHFP = SHFM
      ELSE
         SHFM = 0.D0
         SHFP = 0.D0
      ENDIF
C
      L1 = L + 1
C
      IF (ABS(EPS) .GE. EPSO) THEN
         EPSV = ABS(EPS)
      ELSE
         EPSV = EPSO
      ENDIF
C
      EPSQ = 0.2D0 * SQRT(EPSV)
C
      H = HALF
C
      IS = 2**NPOW
      IM = IS
C
      KM = 0
      KP = 0
      NM = 0
      NP = 0
C
      VNEW = FUNC(X0, Y0, Z0, D0,
     &     AO(L1)*FAC+SHFP, Z, X1, X2) * BO
*
* ---- INITIAL STEP -----
*      INTEGRATE WITH MESH SIZE = 0.5
*      AND CHECK DECAY OF INTEGRAND
*
      DO 10 I = IS, NEND, IM
C
         IF (KM .LE. 1) THEN
            WM = FUNC(X0, Y0, Z0, D0,
     &           AM(I, L1)*FAC + SHFM, Z, X1, X2) * BB(I)
            VNEW = VNEW + WM
            IF (ABS(WM) .LE. EPSV) THEN
               KM = KM + 1
               IF (KM .GE. 2) NM = I - IM
            ELSE
               KM = 0
            ENDIF
         ENDIF
C
         IF (KP .LE. 1) THEN
            WP = FUNC(X0, Y0, Z0, D0,
     &           AP(I, L1)*FAC + SHFP, Z, X1, X2) * BB(I)
            VNEW = VNEW + WP
            IF (ABS(WP) .LE. EPSV) THEN
               KP = KP + 1
               IF (KP .GE. 2) NP = I - IM
            ELSE
               KP = 0
            ENDIF
         ENDIF
C
         IF (KM .EQ. 2 .AND. KP .EQ. 2) GOTO 11
C
   10 CONTINUE
C
   11 CONTINUE
C
      IF (NM .EQ. 0) THEN
         NM = NEND
         EPSM = SQRT (ABS(WM))
      ENDIF
C
      IF (NP .EQ. 0) THEN
         NP = NEND
         EPSP = SQRT (ABS(WP))
      ENDIF
C
      EPSQ = MAX (EPSQ, EPSM, EPSP)
*
* ----- GENERAL STEP -----
*
```

```
      VOLD = H * FAC * VNEW
C
      DO 20 MSTEP = 1, NPOW
C
         VNEW = 0.0
C
         IH = IS
         IS = IS / 2
C
         DO 540 I = IS, NM, IH
            VNEW = VNEW
     &           + FUNC(X0, Y0, Z0, D0,
     &           AM(I, L1)*FAC + SHFM, Z, X1, X2) * BB(I)
  540    CONTINUE
C
         DO 550 I = IS, NP, IH
            VNEW = VNEW
     &           + FUNC(X0, Y0, Z0, D0,
     &           AP(I, L1)*FAC + SHFP, Z, X1, X2) * BB(I)
  550    CONTINUE
C
         VNEW = (VOLD + H * FAC * VNEW) * HALF
C
         IF (ABS(VNEW - VOLD) .LT. EPSQ) THEN
*
* ----- CONVERGED AND RETURN ------
*
            V = VNEW
            RETURN
         ENDIF
C
         H = H * HALF
         VOLD = VNEW
C
   20 CONTINUE
C
      V = VNEW
      RETURN
C
      END
C
C
C
      SUBROUTINE FIAB (NEND, NPOW, AO, AM, AP, BO, BB)
C
      IMPLICIT REAL*8 (A-H, O-Z)
      DIMENSION AM(608,2), AO(2), AP(608,2), BB(608)
C
      PARAMETER (ONE = 1)
      PARAMETER (HALF = ONE / 2)
      PARAMETER (N6 = 6)
      PARAMETER (H = ONE / 2**(N6 + 1))
C
      A9 = 0.9999 9999 9999 9998 D0
*
* ---- START COMPUTATION OF POINTS AND WEIGHTS -----
*
      PH = 2 * ATAN (ONE)
C
      NPOW = N6
      NEND = 608
C
      AO(1) = 0.D0
      AO(2) = 1.D0
      BO    = PH
C
      EH = EXP (H)
      EN = 1.D0
C
      DO 10 I = 1, NEND
         EN = EH * EN
         ENI = 1.D0 / EN
         SH = (EN - ENI) * HALF
         CH = (EN + ENI) * HALF
         EXS = EXP (PH * SH)
         EXSI = 1.D0 / EXS
         CHSI = 2.D0 / (EXS + EXSI)
         AP(I,1) = ((EXS - EXSI) * HALF) * CHSI
         IF (AP(I,1) .GE. A9) AP(I,1) = A9
         AP(I,2) = EXSI * CHSI
         AM(I,1) = - AP(I,1)
         AM(I,2) = - AP(I,2)
         BB(I) = PH * CH * CHSI**2
   10 CONTINUE
C
      AP(608,2) = AP(607,2)
      AM(608,2) = AM(607,2)
C
      RETURN
      END
C
C
C
      DOUBLE PRECISION FUNCTION MGXYX(X0, Y0, Z0, D0, Y, Z, X1, X2)
      IMPLICIT REAL*8 (A-H, O-Z)
      DIMENSION DI(10,10), PHI(10,10), PSI(10,10)
      COMMON /COM2/ C00, C0X, C0Y, C0Z
C
      ZT2=Y*Y+(Z-D0)*(Z-D0)
      CC2=(Y0-Y)*(Y0-Y)+(Z0-Z)*(Z0-Z)
      ZT=SQRT(ZT2)
      CC=SQRT(CC2)
C
      CALL PHSIIJ(X0, ZT, CC, X1, X2, PHI, PSI)
C
      MGXYX=C0X*(3.D0*(Z0-Z)*(X0*PHI(3,5)-PSI(3,5)))
     &     +C0Y*(3.D0*(Z0-Z)*Y*PHI(3,5))
     &     +C0Z*(3.D0*(Z0-Z)*Z*PHI(3,5))
      RETURN
      END
C
C
C
      DOUBLE PRECISION FUNCTION MGXYY(X0, Y0, Z0, D0, Y, Z, X1, X2)
      IMPLICIT REAL*8 (A-H, O-Z)
      DIMENSION DI(10,10), PHI(10,10), PSI(10,10)
      COMMON /COM2/ C00, C0X, C0Y, C0Z
C
      ZT2=Y*Y+(Z-D0)*(Z-D0)
      CC2=(Y0-Y)*(Y0-Y)+(Z0-Z)*(Z0-Z)
      ZT=SQRT(ZT2)
      CC=SQRT(CC2)
C
      CALL PHSIIJ(X0, ZT, CC, X1, X2, PHI, PSI)
C
      MGXYY=C0X*(3.D0*(Z0-Z)*(Y0-Y)*(PHI(3,5)-PSI(3,5)))
     &     +C0Y*(3.D0*(Z0-Z)*Y*(Y0-Y)*PHI(3,5))
     &     +C0Z*(3.D0*(Z0-Z)*Z*(Y0-Y)*PHI(3,5))
      RETURN
      END
C
C
C
      DOUBLE PRECISION FUNCTION MGXYZ(X0, Y0, Z0, D0, Y, Z, X1, X2)
      IMPLICIT REAL*8 (A-H, O-Z)
      DIMENSION DI(10,10), PHI(10,10), PSI(10,10)
      COMMON /COM2/ C00, C0X, C0Y, C0Z
C
```

```
      ZT2=Y*Y+(Z-D0)*(Z-D0)
      CC2=(Y0-Y)*(Y0-Y)+(Z0-Z)*(Z0-Z)
      ZT=SQRT(ZT2)
      CC=SQRT(CC2)
C
      CALL PHSIIJ(X0, ZT, CC, X1, X2, PHI, PSI)
C
      MGXYZ=C0X*(3.D0*(Z0-Z)*(Y0-Y)*(PHI(3,3)-PSI(3,5)))
     &     +C0Y*(3.D0*(Z0-Z)*Y*(Y0-Y)*PHI(3,5))
     &     +C0Z*(3.D0*(PHI(1,1)-(Z0-Z)*(Z0-Z)*PHI(3,5)))
      RETURN
      END
C
C
C
      DOUBLE PRECISION FUNCTION MGXZX(X0, Y0, Z0, D0, Z, Y, X1, X2)
      IMPLICIT REAL*8 (A-H, O-Z)
      DIMENSION DI(10,10), PHI(10,10), PSI(10,10)
      COMMON /COM2/ C00, C0X, C0Y, C0Z
C
      ZT2=Y*Y+(Z-D0)*(Z-D0)
      CC2=(Y0-Y)*(Y0-Y)+(Z0-Z)*(Z0-Z)
      ZT=SQRT(ZT2)
      CC=SQRT(CC2)
C
      CALL PHSIIJ(X0, ZT, CC, X1, X2, PHI, PSI)
C
      MGXZX=C0X*(3.D0*(Y0-Y)*(X0*PHI(3,3)-PSI(3,5)))
     &     +C0Y*(3.D0*(Y0-Y)*Y*PHI(3,5))
     &     +C0Z*(3.D0*(Y0-Y)*Z*PHI(3,5))
      RETURN
      END
C
C
C
      DOUBLE PRECISION FUNCTION MGXZY(X0, Y0, Z0, D0, Z, Y, X1, X2)
      IMPLICIT REAL*8 (A-H, O-Z)
      DIMENSION DI(10,10), PHI(10,10), PSI(10,10)
      COMMON /COM2/ C00, C0X, C0Y, C0Z
C
      ZT2=Y*Y+(Z-D0)*(Z-D0)
      CC2=(Y0-Y)*(Y0-Y)+(Z0-Z)*(Z0-Z)
      ZT=SQRT(ZT2)
      CC=SQRT(CC2)
C
      CALL PHSIIJ(X0, ZT, CC, X1, X2, PHI, PSI)
C
      MGXZY=C0X*(3.D0*(Y0-Y)*(Y0-Y)*(PHI(3,3)-PSI(3,5)))
     &     +C0Y*(3.D0*(Y0-Y)*Y*(Y0-Y)*PHI(3,5))
     &     +C0Z*(3.D0*(Y0-Y)*Z*(Y0-Y)*PHI(3,5))
      RETURN
      END
C
C
C
      DOUBLE PRECISION FUNCTION MGXZZ(X0, Y0, Z0, D0, Z, Y, X1, X2)
      IMPLICIT REAL*8 (A-H, O-Z)
      DIMENSION DI(10,10), PHI(10,10), PSI(10,10)
      COMMON /COM2/ C00, C0X, C0Y, C0Z
C
      ZT2=Y*Y+(Z-D0)*(Z-D0)
      CC2=(Y0-Y)*(Y0-Y)+(Z0-Z)*(Z0-Z)
      ZT=SQRT(ZT2)
      CC=SQRT(CC2)
C
      CALL PHSIIJ(X0, ZT, CC, X1, X2, PHI, PSI)
C
      MGXZZ=C0X*(3.D0*(Y0-Y)*(Y0-Y)*(PHI(3,3)-PSI(3,5)))
     &     +C0Y*(3.D0*Y*(Y0-Y)*(Y0-Y)*PHI(3,5))
     &     +C0Z*(3.D0*(Y0-Y)*PHI(3,3))
      RETURN
      END
C
C
C
      DOUBLE PRECISION FUNCTION MGYZX(X0, Y0, Z0, D0, X, Z, Y1, Y2)
      IMPLICIT REAL*8 (A-H, O-Z)
      DIMENSION DI(10,10), PHI(10,10), PSI(10,10)
      COMMON /COM2/ C00, C0X, C0Y, C0Z
C
      ZT2=X*X+(Z-D0)*(Z-D0)
      CC2=(X0-X)*(X0-X)+(Z0-Z)*(Z0-Z)
      ZT=SQRT(ZT2)
      CC=SQRT(CC2)
C
      CALL PHSIIJ(Y0, ZT, CC, Y1, Y2, PHI, PSI)
C
      MGYZX=C0X*(3.D0*(X0-X)*X*PHI(3,5))
     &     +C0Y*(3.D0*(X0-X)*(X0*PHI(3,3)-PSI(3,5)))
     &     +C0Z*(3.D0*(X0-X)*Z*PHI(3,5))
      RETURN
      END
C
C
C
      DOUBLE PRECISION FUNCTION MGYZY(X0, Y0, Z0, D0, Z, Y, X1, X2)
      IMPLICIT REAL*8 (A-H, O-Z)
      DIMENSION DI(10,10), PHI(10,10), PSI(10,10)
      COMMON /COM2/ C00, C0X, C0Y, C0Z
C
      ZT2=X*X+(Z-D0)*(Z-D0)
      CC2=(X0-X)*(X0-X)+(Z0-Z)*(Z0-Z)
      ZT=SQRT(ZT2)
      CC=SQRT(CC2)
C
      CALL PHSIIJ(Y0, ZT, CC, Y1, Y2, PHI, PSI)
C
      MGYZY=C0X*(3.D0*(X0-X)*(X0-X)*PHI(3,5))
     &     +C0Y*(3.D0*(X0-X)*(Y0*PHI(3,3)-PSI(3,5)))
     &     +C0Z*(3.D0*(X0-X)*Z*(X0-X)*PHI(3,5))
      RETURN
      END
C
C
C
      DOUBLE PRECISION FUNCTION MGYZZ(X0, Y0, Z0, D0, Z, Y, X1, X2)
      IMPLICIT REAL*8 (A-H, O-Z)
      DIMENSION DI(10,10), PHI(10,10), PSI(10,10)
      COMMON /COM2/ C00, C0X, C0Y, C0Z
C
      ZT2=X*X+(Z-D0)*(Z-D0)
      CC2=(X0-X)*(X0-X)+(Z0-Z)*(Z0-Z)
      ZT=SQRT(ZT2)
      CC=SQRT(CC2)
C
      CALL PHSIIJ(Y0, ZT, CC, Y1, Y2, PHI, PSI)
C
      MGYZZ=C0X*(3.D0*X*(X0-X)*(X0-X)*PHI(3,5))
     &     +C0Y*(3.D0*(X0-X)*(Y0*PHI(3,3)-PSI(3,5)))
     &     +C0Z*(3.D0*Z*(Z0-Z)*(X0-X)*PHI(3,5))
      RETURN
      END
C
C
C
      SUBROUTINE PHSIIJ(X0, ZT, CC, X1, X2, PHI, PSI)
      IMPLICIT REAL*8 (A-H,O-Z)
      DIMENSION DI(10,10), PHI(10,10), PSI(10,10)
```

```
C
      ZT2=ZT*ZT
      CC2=CC*CC
C
      CALL PQAB(X0,ZT2,CC2,X1,X2,ALP,BET,PALP,PBET,
     &     QALP,QBET,DK,XI,ET,T1,T2,PH1,PH2)
      CALL ELLIPFE(PH1,DK,ELPF1,ELPE1)
      CALL ELLIPFE(PH2,DK,ELPF2,ELPE2)
      ELPF=ELPF2-ELPF1
      ELPE=ELPE2-ELPE1
C
      XI2=XI*XI
      ET2=ET*ET
      TX1=T1*T1+XI2
      TX2=T2*T2+XI2
      TE1=T1*T1+ET2
      TE2=T2*T2+ET2
C
C     I11
      DI(1,1)=ELPF/XI
C     I13
      DI(1,3)=(XI2*ELPE-ET2*ELPF)/(XI*ET2*(XI2-ET2))
C     I31
      DI(3,1)=(ELPF-ELPE)/(XI*(XI2-ET2))
     &     +T2/(XI2*SQRT(TX2*TE2))-T1/(XI2*SQRT(TX1*TE1))
C     I33
      DI(3,3)=((XI2+ET2)*ELPE-2.D0*ET2*ELPF)/(XI*ET2*
     &     (XI2-ET2)*(XI2-ET2))
     &     +(T2/(XI2*SQRT(TX2*TE2))
     &     -T1/(XI2*SQRT(TX1*TE1)))/(XI2*(XI2-ET2))
C     I51
      DI(5,1)=((3.D0*XI2-ET2)*ELPF
     &     -2.D0*(2.D0*XI2-ET2)*ELPE)/(3.D0*XI2*XI*(XI2-ET2)*(XI2-ET2))
     &     +(3.D0*XI2-2.D0*ET2)*(T2/(XI2*SQRT(TX2*TE2))
     &     -T1/(XI2*SQRT(TX1*TE1)))/(3.D0*XI2*XI2*(XI2-ET2))
     &     +(T2/(XI2*SQRT(TX2*TX2*TX2*TE2))
     &     -T1/(XI2*SQRT(TX1*TX1*TX1*TE1)))/(3.D0*XI2)
C     I35
      DI(3,5)=(DI(3,3)-DI(5,1))/(XI2-ET2)
C
      AB=ALP-BET
C
C     PHI11, PSI11
      PHI(1,1)=AB*DI(1,1)/SQRT(PALP*QALP)
      PSI(1,1)=AB*(BET*DI(1,1)
     &     +0.5D0*ALP*(LOG(XI2+ET2+2.D0*T2*T2+2.D0*SQRT(TE2*TX2))
     &     -LOG(XI2+ET2+2.D0*T1*T1+2.D0*SQRT(TE1*TX1))))/SQRT(PALP*QALP)
C     PHI33
      PHI(3,3)=AB*(DI(1,1)-(XI2+ET2-6.D0)*DI(1,3)
     &     +(XI2*XI2-6.D0*XI2+1.D0)*DI(3,3)
     &     +2.D0*(SQRT(TX2*TE2)
     &     +(XI2-1.D0)/((ET2-XI2)*(ET2-XI2)*TX2)
     &     +(ET2-1.D0)/((ET2-XI2)*(ET2-XI2)*TE2))
     &     -2.D0*(SQRT(TX1*TE1)
     &     +(XI2-1.D0)/((ET2-XI2)*(ET2-XI2)*TX1)
     &     +(ET2-1.D0)/((ET2-XI2)*(ET2-XI2)*TE1)))/(SQRT(PALP*QALP)**3)
C     PHI35 PSI35
      PHI(3,5)=AB*(DI(1,1)-(XI2+2.D0*ET2-15.D0)*DI(3,1)
     &     +3.D0*(ET2*ET2-10.D0*ET2+5.D0)*DI(3,3)
     &     -(ET2**3-15.D0*ET2*ET2+15.D0*ET2-1.D0)*DI(3,5))
      PSI(3,5)=AB*(ALP*DI(1,1)-(ALP*(XI2+2.D0*ET2)
     &     -5.D0*(2.D0*ALP+BET))*DI(3,1)
     &     +(3.D0*ALP*ET2*ET2-10.D0*ET2*(2.D0*ALP+BET)
     &     +5.D0*(ALP+2.D0*BET))*DI(3,3)
     &     -(ALP*ET2**3-5.D0*ET2*ET2*(2.D0*ALP+BET)
     &     +5.D0*ET2*(ALP+2.D0*BET)+BET)*DI(3,5))
      RETURN
      END
C
C
C
      SUBROUTINE PQAB(X0,ZT2,CC2,X1,X2,ALP,BET,PALP,PBET,
     &     QALP,QBET,DK,XI,ET,T1,T2,PH1,PH2)
      IMPLICIT REAL*8 (A-H,O-Z)
      DATA F0,F1/0.D0,1.D0/
C
      X02=X0*X0
      IF(X0.NE.F0) THEN
         DET2=(CC2+X02-ZT2)*(CC2+X02-ZT2)+4.D0*X02*ZT2
         DET=SQRT(DET2)
         ALP = -2.D0 * X0 * ZT2 / (CC2+X02-ZT2+DET)
         BET=(CC2+X02-ZT2+DET)/(2.D0*X0)
         ALP2=ALP*ALP
         BET2=BET*BET
C
         PALP=ALP2+ZT2
         PBET=BET2+ZT2
         QALP=(X0-ALP)*(X0-ALP)+CC2
         QBET=(X0-BET)*(X0-BET)+CC2
C
         T1=(X1-BET)/(ALP-X1)
         T2=(X2-BET)/(ALP-X2)
         XI2=PBET/PALP
         ET2=QBET/QALP
         XI=SQRT(XI2)
         ET=SQRT(ET2)
      ELSE
         T1=X1
         T2=X2
         IF(ZT2.GT.CC2) THEN
            XI2=ZT2
            ET2=CC2
         ELSE
            XI2=CC2
            ET2=ZT2
         ENDIF
         XI=SQRT(XI2)
         ET=SQRT(ET2)
         ALP=F1
         BET=F0
         PALP=F1
         QALP=F1
      ENDIF
      DK2=(XI2-ET2)/XI2
      DK=SQRT(DK2)
      PH1=DATAN(T1/ET)
      PH2=DATAN(T2/ET)
C
      RETURN
      END
C
C
C
      SUBROUTINE ELLIPFE(PH0, G0, ELPF, ELPE)
      IMPLICIT REAL*8 (A-H,O-Z)
      DATA F0,F1,PI,EPS,INF/0.D0,1.D0,3.1415926535897932D0,
     &     2.22044604925031D-16,2.D16/
C
      G=ABS(G0)
      G2=G*G
      PH=PH0
C
      IF(G.EQ.F1) THEN
         ELPE = DSIN(PH)
         IF(ABS(PH0).LT.PI) ELPF=DLOG((F1+DSIN(PH))/DCOS(PH))
         IF(ABS(PH0).GE.PI) ELPF=INF
         GOTO 300
      ENDIF
```

```
C
      A0=F1
      B0=DSQRT(F1-G2)
      P=F1
      ARG=F0
      CE1=F1
      CE2=F0
  100 A = 0.5D0*(A0+B0)
      B = DSQRT(A0*B0)
      C = 0.5D0*(A0-B0)
      CE1 = CE1-(2.D0**P)*(A*C)
      IF(PH0.NE.90.D0)  THEN
         ARG = REAL(NINT(PH/PI))
         IF(PH.GT.F0.AND.MOD(PH/PI+0.5D0,F1).EQ.F0) ARG=ARG-F1
         PH = PH+DATAN(B0*DTAN(PH)/A0)+PI*ARG
         CE2 = CE2+C*DSIN(PH)
      ENDIF
C
      IF(ABS(A-A0).LE.EPS.AND.ABS(B-B0).LE.EPS.
     &     AND.ABS(A-B).LE.EPS) THEN
         GOTO 200
      ELSE
         A0 = A
         B0 = B
         P = P+F1
         GOTO 100
      ENDIF
C
  200 IF(PH0.NE.PI) AGMPH = PH/(2.D0**P)
      IF(PH0.EQ.PI) AGMPH = PI/2.D0
      ELPF = AGMPH/A
      ELPE = CE1*AGMPH/A+CE2
  300 RETURN
      END
```